ZAŁĄCZNIK C


r0=dm(ile_zliczen);

r1=1230;

COMP(R0,R1);

IF GE JUMP faza_konc;

r0=r0+1;//kolejna probka

dm(ile_zliczen)=r0;

//obliczenie fazy pierwszej

r1=-31;

r0=dm(Right_Channel_In1);

F3=float r0 by r1; //wartosc odebranej probki jako float

M0=1;

M1=0;

I0=podst_Q1;

I1=podst_Q2;

I2=coefp;

//osiem podstawowych czestotliwosc

LCNTR=8, DO aktual_podst UNTIL lce;


F0=dm(I0,M1); //Qk(n-1)


F1=dm(I1,M1); //Qk(n-2)


F2=dm(I2,M0);//coef


F2=F0*F2;//COEF*Qk(n-1)


F1=F1-F3,DM(I1,M0)=F0;//Qk(n-2)-x(n)


F2=F2-F1;

 
DM(I0,M0)=F2;

aktual_podst: nop;nop;

jump koniec_procedury;

faza_konc:

r0=0;

dm(ile_zliczen)=r0; /*sample_count = 0;*/

M0=1;

M1=0;

I0=podst_Q1;

I1=podst_Q2;

I2=coefp;

//obliczenie wartości mocy
LCNTR=8, DO war_moc UNTIL lce;


F0=dm(I0,M1); //Qk(n-1)


f8=f0;


F1=dm(I1,M0); //Qk(n-2)


f9=f1;


F2=dm(I2,M0);//coef


F2=F0*F2;//COEF*Qk(n-1)


F2=F2*F1;


F0=F0*F8;//Qk(n-1)^2


F1=F1*F9;//Qk(n-2)^2


F0=F0+F1;//Qk(n-2)^2 + Qk(n-1)^2


F0=F0-F2; //Yk(n)

 
DM(I0,M0)=F0;

war_moc: nop;nop;
//detekcja wartości maksymalnych
M0=1;

M1=4;

I0=podst_Q1;

I1=podst_Q1;

F0=dm(I0, M0); //F0 wartosc pierwszego wiersz; 

R2=0;//R2 - numer wiersza o max mocy

F4=dm(I1, M1); //F4 wartosc pierwszej kolumny;

F4=dm(I1, M0); //F4 wartosc pierwszej kolumny;

R6=0;//R6 - numer kolumny o max mocy

R7=0;//LICZNIK

LCNTR=3, DO max_war UNTIL lce;



r7=r7+1; //ikrementacja aktualnego numeru wiersza i kolumny



F5=DM(I1,M0);



COMP(F5,F4); //CZY ZMIENIONO WARTOSC MOCY



IF LE JUMP nie_kol;



F4=F5;



R6=R7;

nie_kol:



F1=DM(I0,M0);



COMP(F1,F0); //CZY ZMIENIONO WARTOSC MOCY



IF LE JUMP max_war;



F0=F1;



R2=R7;

max_war: nop;nop;
F1=3.000;

COMP(F0,F1);

IF LE JUMP nie_wykryto;

COMP(F4,F1);

IF LE JUMP nie_wykryto;

F1=0.015;

F1=F1*F4;

COMP(F0,F1);

IF LT JUMP nie_wykryto;// Sprawdzenie "Normal twist"

F1=0.015;

F1=F1*F0;

COMP(F4,F1);

IF LT JUMP nie_wykryto;// Sprawdzenie "Reverse twist"

F1=0.02;

F1=F1*F0;

COMP(F0,F4);

IF GE JUMP czy_druga_czest;

F1=F1*F4;

czy_druga_czest:

M0=1;

I0=podst_Q1;

F5=0;

LCNTR=8, DO spr_2_czest UNTIL lce;//druga czestotliwosc o wysokim poziomie


F3=dm(I0,M0);


COMP(F1,F3);


IF GE JUMP wszyst_OK;


F3=1.000;


F5=F5+F3;

wszyst_OK:

spr_2_czest: nop;nop;


f1=2.000;

comp(f5,f1);

if GT jump nie_wykryto;

//wyswietlamy liczbe DTMF

I0=liczba_DTMF;

R2=LSHIFT R2 BY 2;

R2=R2+R6; //numer wiersza razy 4 plus numer kolumny 

m0=r2;

wys:

ustat2=dm(M0,I0);

dm(IOFLAG)=ustat2; //zapalenie numeru diody

//funkcja opoźnienia

lcntr=6250000;

do opdelay until lce;

opdelay: nop;

bit set ustat2 FLG9O|FLG8O|FLG7O|FLG6O|FLG5O|FLG4O;

bit clr ustat2 FLG9|FLG8|FLG7|FLG6|FLG5|FLG4;



dm(IOFLAG)=ustat2; //zapalenie numeru diody


nie_wykryto: 

//zerujemy wartosci poczatkowe Q1 i Q2

M0=1;

I0=podst_Q1;

I1=podst_Q2;

r7=0;

LCNTR=8, DO zerowanie UNTIL lce;

 
DM(I0,M0)=r7;

 
DM(I1,M0)=r7;

zerowanie: nop;nop;

koniec_procedury:

